InFoMM Continuum Methods for Industry ①

Exercises #2

1. Scale 2-D Navier Stokes eq^{n}s as suggested to get

$$\begin{cases} \dfrac{\partial u}{\partial x} + \dfrac{\partial v}{\partial y} = 0 \\ R^* \left[\dfrac{\partial u}{\partial t} + u \dfrac{\partial u}{\partial x} + v \dfrac{\partial u}{\partial y} \right] = -\dfrac{\partial p}{\partial x} + \varepsilon^2 \dfrac{\partial^2 u}{\partial x^2} + \dfrac{\partial^2 u}{\partial y^2} \\ \varepsilon^2 R^* \left[\dfrac{\partial v}{\partial t} + u \dfrac{\partial v}{\partial x} + v \dfrac{\partial v}{\partial y} \right] = -\dfrac{\partial p}{\partial y} + \varepsilon^4 \dfrac{\partial^2 v}{\partial x^2} + \varepsilon^2 \dfrac{\partial^2 v}{\partial y^2} \end{cases}$$

BCs $\quad V = \dfrac{\partial u}{\partial y} = 0 \quad$ at $y = 0$

& under the roller, the top surface is given by

$$h_0 h = h_m + R - \sqrt{R^2 - \frac{h_0^2 x^2}{\varepsilon^2}}$$ in dimensionless variables

recall $\dfrac{R}{h_0} = 1/\varepsilon^2$, $\dfrac{h_m}{h_0} = 1$

$$\therefore h(x) = 1 + \frac{1}{\varepsilon^2}\left[1 - \sqrt{1 - \varepsilon^2 x^2}\right]$$

$$\therefore \boxed{h(x) = 1 + \tfrac{1}{2}x^2 + O(\varepsilon^2)}$$

& BCs are $v = u\,h'(x)$, $u + \varepsilon^2 h'(x) v = \gamma\sqrt{1+\varepsilon^2 h'(x)^2}$ ② at $y = h(x)$

Now neglect terms of order ε^2 and R^* to get

$$\boxed{\begin{aligned} \frac{\partial u}{\partial x} + \frac{\partial v}{\partial y} &= 0 \\ \frac{\partial p}{\partial x} &= \frac{\partial^2 u}{\partial y^2} \\ \frac{\partial p}{\partial y} &= 0 \end{aligned}}$$

& $$\boxed{\begin{aligned} v = \frac{\partial u}{\partial y} = 0 \quad &\text{at } y = 0 \\ u = \gamma,\ v = \gamma h'(x) \quad &\text{at } y = h(x) \end{aligned}} \qquad (2.11)$$

(2.10): $$\gamma \int_{-\alpha}^{\beta} \frac{dx}{\left(\eta + \frac{1}{2}x^2\right)^2} = \int_{-\alpha}^{\beta} \frac{dx}{\left(\eta + \frac{1}{2}x^2\right)^3}$$

Let $x = \sqrt{2\eta}\, X$

$$\gamma\eta \int_{-\alpha/\sqrt{2\eta}}^{\beta/\sqrt{2\eta}} \frac{dX}{\left(1+X^2\right)^2} = \int_{-\alpha/\sqrt{2\eta}}^{\beta/\sqrt{2\eta}} \frac{dX}{\left(1+X^2\right)^3}$$

Recall $\alpha = \sqrt{2(1-\eta)}$

3

$$\therefore \boxed{\gamma = \frac{F_2\left(\beta/\sqrt{2\eta}\right) + F_2\left(\sqrt{\frac{1-\eta}{\eta}}\right)}{\eta\left[F_1\left(\frac{\beta}{\sqrt{2\eta}}\right) + F_2\left(\sqrt{\frac{1-\eta}{\eta}}\right)\right]}}$$

with F_1 & F_2 as given.

In[1]:= Integrate[1 / (1 + x^2) ^ 2, x]

Out[1]= $\frac{1}{2}\left(\frac{x}{1+x^2}+\text{ArcTan}[x]\right)$

In[2]:= F1[x_] := $\frac{1}{2}\left(\frac{x}{1+x^2}+\text{ArcTan}[x]\right)$

In[3]:= Integrate[1 / (1 + x^2) ^ 3, x]

Out[3]= $\frac{1}{8}\left(\frac{x\,(5+3\,x^2)}{(1+x^2)^2}+3\,\text{ArcTan}[x]\right)$

In[4]:= F2[x_] := $\frac{1}{8}\left(\frac{x\,(5+3\,x^2)}{(1+x^2)^2}+3\,\text{ArcTan}[x]\right)$

```
In[5]:= ga[b_, y_] := (F2[b / Sqrt[2 y]] + F2[Sqrt[1 - y] / Sqrt[y]]) / y /
          (F1[b / Sqrt[2 y]] + F1[Sqrt[1 - y] / Sqrt[y]])

In[6]:= pl[y_] := Graphics[Text[y, {Sqrt[2 (1 - y)], ga[0, y] - 0.3}]]

In[7]:= pp[y_] := Plot[Evaluate[ga[b, y]], {b, 0, Sqrt[2 (1 - y)]}]

In[8]:= Table[pp[y], {y, 0.1, 0.9, 0.1}];

In[9]:= Table[pl[y], {y, 0.1, 0.9, 0.1}];

In[10]:= Show[%%, %, PlotRange → All, AxesOrigin → {0, 0}, AxesLabel → {β, γ}]
```

Out[10]=



In[11]:= (*γ versus β for different values of η. Note that the value of γ at β= α is the same as that at β=0.
In each case there is a range of values of γ for which there are two positive roots for β.*)

In[12]:= (* Now let η=1-ϵ, β=ϵ^{1/2}b and γ=1+ϵΓ/3, where ϵ<<1.*)

```
In[13]:= Series[Evaluate[ga[Sqrt[e] b, 1 - e]], {e, 0, 1}] // FullSimplify
```

Out[13]= $1+\frac{1}{6}\left(4+\left(\sqrt{2}-b\right)b\right)e+O[e]^{3/2}$

5

In[14]:= `% - 1 - e Γ / 3`

Out[14]= $\left(\frac{1}{6}\left(4+\left(\sqrt{2}-b\right)b\right)-\frac{\Gamma}{3}\right)e+O[e]^{3/2}$

In[15]:= `Normal[% / e]`

Out[15]= $\frac{1}{6}\left(4+\left(\sqrt{2}-b\right)b\right)-\frac{\Gamma}{3}$

In[16]:= `Solve[% == 0, b]`

Out[16]= $\left\{\left\{b\to\frac{1}{\sqrt{2}}-\frac{\sqrt{9-4\,\Gamma}}{\sqrt{2}}\right\},\left\{b\to\frac{1}{\sqrt{2}}+\frac{\sqrt{9-4\,\Gamma}}{\sqrt{2}}\right\}\right\}$

In[17]:= `b /. %[[1]]`

Out[17]= $\frac{1}{\sqrt{2}}-\frac{\sqrt{9-4\,\Gamma}}{\sqrt{2}}$

In[18]:= (*The smaller root is the physically relevant solution. We need Γ>2 so that this root is positive, and Γ<9/4 so that it is real*)

2. Start from

$$\frac{\partial u}{\partial x} + \frac{\partial v}{\partial y} = 0 \, , \quad \frac{\partial p}{\partial x} = \frac{\partial}{\partial y}\left(\mu \frac{\partial u}{\partial y}\right) \, , \quad \frac{\partial p}{\partial y} = 0$$

where now

$$\mu = K\left[2\left(\frac{\partial u}{\partial x}\right)^2 + 2\left(\frac{\partial v}{\partial y}\right)^2 + \left(\frac{\partial u}{\partial y} + \frac{\partial v}{\partial x}\right)^2\right]^{\frac{n-1}{2}}$$

Non-dimensionalize $\quad \underline{u} = U(\tilde{u}, \varepsilon \tilde{v})$

$$(x, y) = h_0(\varepsilon^{-1}\tilde{x}, \tilde{y})$$

Then drop tildes. Note that $\frac{\partial u}{\partial y}$ is much greater than any other velocity gradient component (by a factor of ε^{-2} at least, so to lowest order,

$$\mu \sim K\left(\frac{U}{h_0}\right)^{n-1}\left|\frac{\partial u}{\partial y}\right|^{n-1}$$

so define $\boxed{\mu_0 = K\left(\frac{U}{h_0}\right)^{n-1}}$

and non-dimensionalize pressure with

$$p = p_a + \frac{\mu_0 U}{\varepsilon h_0}\hat{p} = p_a + \frac{K}{\varepsilon}\left(\frac{U}{h_0}\right)^n \hat{p}$$

to get (with tildes dropped)

$$\underline{\underline{\frac{\partial u}{\partial x} + \frac{\partial v}{\partial y} = 0 \ (1), \quad \frac{\partial p}{\partial y} = 0, \quad \frac{dp}{dx} = \frac{\partial}{\partial y}\left(\left| \frac{\partial u}{\partial y} \right|^{n-1} \frac{\partial u}{\partial y} \right)}}$$

with BCs (as before)

$$\boxed{\begin{aligned} &v = \frac{\partial u}{\partial y} = 0 \quad \text{at} \quad y = 0 \\ &u = \gamma,\ v = \gamma h'(x) \quad \text{at} \quad y = h(x) \end{aligned}}$$

where $\underline{h(x) = \eta + \tfrac{1}{2}x^2}$ to lowest order.

By integrating (1) w.r.t. y, we get net mass conservation as before:

$$\boxed{\int_0^h u \, dy \equiv 1}$$

Now solve for u: $\left| \frac{\partial u}{\partial y} \right|^{n-1} \frac{\partial u}{\partial y} = \frac{dp}{dx} y$

[satisfying BC at $y = 0$.]

Take $|\,|$ of both sides:

$$\left| \frac{\partial u}{\partial y} \right|^{n} = \left| \frac{dp}{dx} \right| |y| \Rightarrow \left| \frac{\partial u}{\partial y} \right| = \left| \frac{dp}{dx} \right|^{1/n} |y|^{1/n}$$

$\therefore$ in $y \geq 0$,

$$\boxed{\frac{\partial u}{\partial y} = \frac{dp}{dx}\left|\frac{dp}{dx}\right|^{\frac{1}{n}-1} y^{1/n}}$$

with $u = \gamma$ on $y = h$

$$\Rightarrow \boxed{u = \gamma - \frac{dp}{dx}\left|\frac{dp}{dx}\right|^{\frac{1}{n}-1}\left(\frac{n}{n+1}\right)\left[h^{\frac{n+1}{n}} - y^{\frac{1+n}{n}}\right]}$$

$$\therefore \int_0^h u\,dy = \gamma h - \frac{dp}{dx}\left|\frac{dp}{dx}\right|^{\frac{1}{n}-1}\left(\frac{n}{n+1}\right)\left(\frac{n+1}{2n+1}\right) h^{\frac{2n+1}{n}}$$

$$\therefore \boxed{\gamma h - \frac{n\,h^{\frac{2n+1}{n}}}{2n+1}\frac{dp}{dx}\left|\frac{dp}{dx}\right|^{\frac{1}{n}-1} = 1}$$

with $\underline{p(-\alpha) = p(\beta) = 0}$

$$\therefore \underline{\frac{n}{2n+1}\frac{dp}{dx}\left|\frac{dp}{dx}\right|^{\frac{1}{n}-1} = \frac{\gamma}{h^{1+1/n}} - \frac{1}{h^{2+1/n}}}$$

$$\therefore \frac{n}{2n+1}\left|\frac{dp}{dx}\right|^{\frac{1}{n}} = |RHS|$$

$$\left(\frac{n}{2n+1}\right)^n \left|\frac{dp}{dx}\right| = |RHS|^n$$

$$\therefore \left(\frac{n}{2n+1}\right)^n \frac{dp}{dx} = RHS\,|RHS|^{n-1}$$

$$\therefore \boxed{\int_{-\alpha}^{\beta}\left(\frac{\gamma}{h^{1+1/n}} - \frac{1}{h^{2+1/n}}\right)\left|\frac{\gamma}{h^{1+1/n}} - \frac{1}{h^{2+1/n}}\right|^{n-1} dx = 0}$$

3 $\quad \left(A - (\lambda + \varepsilon) I\right) \underset{\sim}{x} = \underset{\sim}{b}$

where λ is an eigenvalue of A.

Try $\underset{\sim}{x} \sim \underset{\sim}{x}_0 + \varepsilon \underset{\sim}{x}_1 + \ldots$

to get at lowest order

$$\left(A - \lambda I\right) \underset{\sim}{x}_0 = \underset{\sim}{b}$$

But $\det (A - \lambda I) = 0$ so this generically has no solution.

Instead, let $\underset{\sim}{x} = \frac{1}{\varepsilon} \underset{\sim}{X}$ to get

$$\boxed{\left(A - (\lambda + \varepsilon) I\right) \underset{\sim}{X} = \varepsilon \underset{\sim}{b}}$$

$\underset{\sim}{X} \sim \underset{\sim}{X}_0 + \varepsilon \underset{\sim}{X}_1 + \ldots$ gives

$$\left(A - \lambda I\right) \underset{\sim}{X}_0 = \underset{\sim}{0}$$

$$\left(A - \lambda I\right) \underset{\sim}{X}_1 = \underset{\sim}{b} + \underset{\sim}{X}_0$$

etc...

Now since $\det (A - \lambda I) = 0$, there are non trivial solutions for X_0 of the form...

$$\boxed{\underset{\sim}{x}_0 = \alpha \underset{\sim}{v}}$$

where $\underset{\sim}{v}$ is the eigenvector of A with eigenvalue λ (wlog normalized so that $|\underset{\sim}{v}| = 1$), and the amplitude α is arbitrary.

At $O(\varepsilon)$ we get

$$(A - \lambda I)\underset{\sim}{x}_1 = \underset{\sim}{b} + \alpha \underset{\sim}{v}$$

But $\det(A - \lambda I) = 0 \Rightarrow$ this has no solutions for $\underset{\sim}{x}_1$ unless the RHS is orthogonal to everything in the kernel of $A^a - \lambda I$.

Here A is real symmetric, so $A^a \equiv A$ & we need

$$(\underset{\sim}{b} + \alpha \underset{\sim}{v}) \cdot \underset{\sim}{v} = 0$$

for solutions for $\underset{\sim}{x}_1$ to exist. This solvability condition determines the amplitude of the leading-order solution:

$$\boxed{\alpha = - \underset{\sim}{b} \cdot \underset{\sim}{v}}$$

4. Given
$$\begin{cases} \varepsilon^2 St\, r \dfrac{\partial u}{\partial t} = \dfrac{\partial}{\partial r}\left(r \dfrac{\partial u}{\partial r}\right) + \varepsilon^2 r \dfrac{\partial^2 u}{\partial z^2} \\ \dfrac{\partial u}{\partial r} - \varepsilon^2 R'(z) \dfrac{\partial u}{\partial z} = -\varepsilon^2 u \sqrt{1+\varepsilon^2 R'(z)^2} \\ \qquad \text{at } r = R(z) \end{cases}$$

Integrate w.r.t. r:
$$\varepsilon^2 St \int_0^R r \frac{\partial u}{\partial t}\, dr = \left[r \frac{\partial u}{\partial r}\right]_0^R + \varepsilon^2 \int_0^R r \frac{\partial^2 u}{\partial z^2}\, dr$$

$$\therefore \cancel{\varepsilon^2} St \frac{\partial}{\partial t} \int_0^R u r\, dr = R \cancel{\varepsilon^2} \left[R' \frac{\partial u}{\partial z} - u\sqrt{1+\varepsilon^2 R'^2}\right]_{r=R} + \cancel{\varepsilon^2} \int_0^R r \frac{\partial^2 u}{\partial z^2}\, dr$$

$$\therefore St \frac{\partial}{\partial t} \int_0^R u r\, dr = \frac{\partial}{\partial z} \int_0^R \frac{\partial u}{\partial z} r\, dr - R u \sqrt{1+\varepsilon^2 R'^2}\Big|_{r=R}$$

using Leibnitz rule: $$\frac{\partial}{\partial z} \int_0^{R(z)} f(r,z,t)\, dr$$
$$= \int_0^R \frac{\partial f}{\partial z}\, dr + R'(z) f(r,z,t)\Big|_{r=R}$$

S_{11}

```
In[1]:= R[z_] := R1 z + R2 z^2 + R3 z^3 + R4 z^4 + R5 z^5

In[2]:= u[z_, t_] := 1 + t f1[z / t] + t^2 f2[z / t] + t^3 f3[z / t] + t^4 f4[z / t] + t^5 f5[z / t]

In[3]:= s[t_] := s1 t + s2 t^2 + s3 t^3 + s4 t^4 + s5 t^5 (St D[u[z, t], t] -
          D[u[z, t], z, z] - 2 R'[z] / R[z] D[u[z, t], z] + 2 u[z, t] / R[z]) /. z → t y

In[4]:= (St D[u[z, t], t] - D[u[z, t], z, z] - 2 R'[z] / R[z] D[u[z, t], z] + 2 u[z, t] / R[z]) /.
          z → t y;

In[5]:= Series[%, {t, 0, 3}] // Simplify
```

Out[5]=

$$\begin{aligned}
&\frac{2 - 2\, R1\, f1'[y] - R1\, y\, f1''[y]}{R1\, y\, t} - \frac{1}{R1^2\, y}\\
&\left(2\, R2\, y - R1\, (2 + R1\, St\, y)\, f1[y] + R1\, y\, (2\, R2 + R1\, St\, y)\, f1'[y] + 2\, R1^2\, f2'[y] + R1^2\, y\, f2''[y]\right) +\\
&\left(\frac{2\, R2^2\, y}{R1^3} - \frac{2\, R3\, y}{R1^2} - \frac{2\, R2\, f1[y]}{R1^2} + 2\left(St + \frac{1}{R1\, y}\right) f2[y] + \frac{2\, R2^2\, y\, f1'[y]}{R1^2} - \right.\\
&\quad \left.\frac{4\, R3\, y\, f1'[y]}{R1} - \frac{2\, R2\, f2'[y]}{R1} - St\, y\, f2'[y] - \frac{2\, f3'[y]}{y} - f3''[y]\right) t + \left(3\, St\, f3[y] - \frac{1}{R1^4}\right.\\
&\quad 2\left(\left(R2^3 - 2\, R1\, R2\, R3 + R1^2\, R4\right) y^2 + R1\left(-R2^2 + R1\, R3\right) y\, f1[y] + R1^2\, R2\, f2[y] - \frac{R1^3\, f3[y]}{y}\right) -\\
&\quad St\, y\, f3'[y] - \frac{1}{R1^3\, y} 2\left(\left(R2^3 - 3\, R1\, R2\, R3 + 3\, R1^2\, R4\right) y^3\, f1'[y] + \right.\\
&\qquad \left.\left. R1\left(-\left(R2^2 - 2\, R1\, R3\right) y^2\, f2'[y] + R1\, (R2\, y\, f3'[y] + R1\, f4'[y])\right)\right) - f4''[y]\right) t^2 +\\
&\left(4\, St\, f4[y] + \frac{1}{R1^5} 2\left(\left(R2^4 - 3\, R1\, R2^2\, R3 + 2\, R1^2\, R2\, R4 + R1^2\left(R3^2 - R1\, R5\right)\right) y^3 - R1\left(R2^3 - 2\, R1\, R2\, R3 + \right.\right.\right.\\
&\qquad \left.\left. R1^2\, R4\right) y^2\, f1[y] - R1^2\left(-R2^2 + R1\, R3\right) y\, f2[y] - R1^3\, R2\, f3[y] + \frac{R1^4\, f4[y]}{y}\right) -\\
&\quad St\, y\, f4'[y] - \frac{1}{R1^4\, y} 2\left(-\left(R2^4 - 4\, R1\, R2^2\, R3 + 4\, R1^2\, R2\, R4 + 2\, R1^2\left(R3^2 - 2\, R1\, R5\right)\right) y^4\, f1'[y] + \right.\\
&\qquad R1\left(\left(R2^3 - 3\, R1\, R2\, R3 + 3\, R1^2\, R4\right) y^3\, f2'[y] + \right.\\
&\qquad \left.\left.\left. R1\left(-\left(R2^2 - 2\, R1\, R3\right) y^2\, f3'[y] + R1\, (R2\, y\, f4'[y] + R1\, f5'[y])\right)\right)\right) - f5''[y]\right) t^3 + O[t]^4
\end{aligned}$$

```
In[6]:= eq = Simplify[% t];

In[7]:= u[s[t], t] - 1 // Simplify;
```

13

In[8]:=
```
Series[%, {t, 0, 5}] // Simplify
```

Out[8]=

$$
\begin{aligned}
&\text{f1}[\text{s1}]\, t + (\text{f2}[\text{s1}] + \text{s2}\,\text{f1}'[\text{s1}])\, t^2 + \left(\text{f3}[\text{s1}] + \text{s3}\,\text{f1}'[\text{s1}] + \text{s2}\,\text{f2}'[\text{s1}] + \frac{1}{2}\,\text{s2}^2\,\text{f1}''[\text{s1}]\right) t^3 + \\
&\quad \Bigg(\text{f4}[\text{s1}] + \text{s3}\,\text{f2}'[\text{s1}] + \text{s2}\,\text{f3}'[\text{s1}] + \text{s2}\,\text{s3}\,\text{f1}''[\text{s1}] + \\
&\qquad \text{f1}'[\text{s1}]\left(\text{s4} + \frac{2\,\text{s5}}{\text{R1}\,\text{y}} - \frac{2\,\text{s5}\,\text{f1}'[\text{y}]}{\text{y}} - \text{s5}\,\text{f1}''[\text{y}]\right) + \frac{1}{2}\,\text{s2}^2\,\text{f2}''[\text{s1}] + \frac{1}{6}\,\text{s2}^3\,\text{f1}^{(3)}[\text{s1}]\Bigg) t^4 + \\
&\quad \Bigg(\text{f5}[\text{s1}] + \text{s3}\,\text{f3}'[\text{s1}] + \text{s2}\,\text{f4}'[\text{s1}] + \text{f2}'[\text{s1}]\left(\text{s4} + \frac{2\,\text{s5}}{\text{R1}\,\text{y}} - \frac{2\,\text{s5}\,\text{f1}'[\text{y}]}{\text{y}} - \text{s5}\,\text{f1}''[\text{y}]\right) + \\
&\qquad \frac{1}{2}\,\text{f1}''[\text{s1}]\left(\text{s3}^2 + 2\,\text{s2}\left(\text{s4} + \frac{2\,\text{s5}}{\text{R1}\,\text{y}} - \frac{2\,\text{s5}\,\text{f1}'[\text{y}]}{\text{y}} - \text{s5}\,\text{f1}''[\text{y}]\right)\right) + \\
&\qquad \text{s2}\,\text{s3}\,\text{f2}''[\text{s1}] - \frac{1}{\text{R1}^2\,\text{y}}\text{s5}\,\text{f1}'[\text{s1}]\,\big(2\,\text{R2}\,\text{y} - \text{R1}\,(2 + \text{R1}\,\text{St}\,\text{y})\,\text{f1}[\text{y}] + \\
&\qquad\quad \text{R1}\,\text{y}\,(2\,\text{R2} + \text{R1}\,\text{St}\,\text{y})\,\text{f1}'[\text{y}] + 2\,\text{R1}^2\,\text{f2}'[\text{y}] + \text{R1}^2\,\text{y}\,\text{f2}''[\text{y}]\big) + \frac{1}{2}\,\text{s2}^2\,\text{f3}''[\text{s1}] + \\
&\qquad \frac{1}{2}\,\text{s2}^2\,\text{s3}\,\text{f1}^{(3)}[\text{s1}] + \frac{1}{6}\,\text{s2}^3\,\text{f2}^{(3)}[\text{s1}] + \frac{1}{24}\,\text{s2}^4\,\text{f1}^{(4)}[\text{s1}]\Bigg) t^5 + O[t]^6
\end{aligned}
$$

In[9]:=
```
bc1 = Simplify[% / t];
```

In[10]:=
```
(D[u[z, t], z] - 1 - s'[t] /. z → s[t]) // Simplify;
```

In[11]:= **Series[%, {t, 0, 4}] // Simplify**

Out[11]=

$$
\begin{aligned}
&(-1 - s1 + f1'[s1]) + (f2'[s1] + s2\,(-2 + f1''[s1]))\,t + \\
&\left(-3\,s3 + f3'[s1] + s3\,f1''[s1] + s2\,f2''[s1] + \frac{1}{2}\,s2^2\,f1^{(3)}[s1]\right) t^2 + \\
&\left(-4\,s4 - \frac{8\,s5}{R1\,y} + f4'[s1] - \frac{2\,s5\,f1'[y]\,(-4 + f1''[s1])}{y} + s4\,f1''[s1] + \right. \\
&\quad \frac{2\,s5\,f1''[s1]}{R1\,y} + 4\,s5\,f1''[y] - s5\,f1''[s1]\,f1''[y] + s3\,f2''[s1] + \\
&\quad \left. s2\,f3''[s1] + s2\,s3\,f1^{(3)}[s1] + \frac{1}{2}\,s2^2\,f2^{(3)}[s1] + \frac{1}{6}\,s2^3\,f1^{(4)}[s1]\right) t^3 + \\
&\left(\frac{10\,R2\,s5}{R1^2} + \frac{10\,s5\,f2'[y]}{y} + f5'[s1] + \frac{1}{R1\,y} s5\,(2 + R1\,St\,y)\,f1[y]\,(-5 + f1''[s1]) - \right. \\
&\quad \frac{2\,R2\,s5\,f1''[s1]}{R1^2} - \frac{2\,s5\,f2'[y]\,f1''[s1]}{y} + s4\,f2''[s1] + \frac{2\,s5\,f2''[s1]}{R1\,y} - s5\,f1''[y]\,f2''[s1] + \\
&\quad 5\,s5\,f2''[y] - s5\,f1''[s1]\,f2''[y] + s3\,f3''[s1] + s2\,f4''[s1] + \frac{1}{2}\,s3^2\,f1^{(3)}[s1] + \\
&\quad s2\,s4\,f1^{(3)}[s1] + \frac{2\,s2\,s5\,f1^{(3)}[s1]}{R1\,y} - s2\,s5\,f1''[y]\,f1^{(3)}[s1] - \frac{1}{R1\,y} s5\,f1'[y] \\
&\quad\ \left(-10\,R2\,y - 5\,R1\,St\,y^2 + y\,(2\,R2 + R1\,St\,y)\,f1''[s1] + 2\,R1\,f2''[s1] + 2\,R1\,s2\,f1^{(3)}[s1]\right) + \\
&\quad s2\,s3\,f2^{(3)}[s1] + \frac{1}{2}\,s2^2\,f3^{(3)}[s1] + \frac{1}{2}\,s2^2\,s3\,f1^{(4)}[s1] + \\
&\quad \left. \frac{1}{6}\,s2^3\,f2^{(4)}[s1] + \frac{1}{24}\,s2^4\,f1^{(5)}[s1]\right) t^4 + O[t]^5
\end{aligned}
$$

In[12]:= **bc2 = Simplify[%];**

In[13]:= **Simplify[eq /. t → 0]**

Out[13]=

$$\frac{2 - 2\,R1\,f1'[y] - R1\,y\,f1''[y]}{R1\,y}$$

In[14]:= **DSolve[% == 0, f1, y]**

Out[14]=

$$\left\{\left\{f1 \to \mathrm{Function}\left[\{y\},\ \frac{y}{R1} - \frac{C[1]}{y} + C[2]\right]\right\}\right\}$$

In[15]:= **f1[y] /. %[[1]] /. C[1] → 0**

Out[15]=

$$\frac{y}{R1} + C[2]$$

In[16]:= **f1[y_] = %**

Out[16]=

$$\frac{y}{R1} + C[2]$$

In[17]:= **(* Must be bounded as y→0 *)**

15

```
In[18]:= {bc1, bc2} /. t → 0
```

Out[18]= $\left\{\frac{s1}{R1}+C[2],\ -1+\frac{1}{R1}-s1\right\}$

```
In[19]:= Solve[% == 0, {C[2], s1}] // Simplify
```

Out[19]= $\left\{\left\{C[2]\to\frac{-1+R1}{R1^2},\ s1\to -1+\frac{1}{R1}\right\}\right\}$

```
In[20]:= s1 = (s1 /. %[[1]]); f1[y_] = (Simplify[f1[y] /. %[[1]]]);

In[21]:= {eq, bc1, bc2} = Simplify[{eq, bc1, bc2} / t];

In[22]:= (* Now repeat procedure *)

In[23]:= Simplify[eq /. t → 0];

In[24]:= DSolve[% == 0, f2, y]
```

Out[24]= $\left\{\left\{f2\to \mathrm{Function}\left[\{y\},\ \frac{1}{3\,R1^3}\left(3\,(-1+R1)\,y+\frac{1}{2}R1\,(2-4\,R2-St+R1\,St)\,y^2-\frac{3\,R1^3\,C[1]}{y}\right)+C[2]\right]\right\}\right\}$

```
In[25]:= f2[y] /. %[[1]] /. C[1] → 0 // Simplify
```

Out[25]= $\frac{1}{6\,R1^3}y\left(-6+R1^2\,St\,y+R1\,(6-(-2+4\,R2+St)\,y)\right)+C[2]$

```
In[26]:= f2[y_] = %;

In[27]:= {bc1, bc2} /. t → 0 // Simplify;

In[28]:= Solve[% == 0, {C[2], s2}] // Simplify;

In[29]:= s2 = (s2 /. %[[1]]); f2[y_] = (Simplify[f2[y] /. %[[1]]]);

In[30]:= {eq, bc1, bc2} = Simplify[{eq, bc1, bc2} / t];

In[31]:= Simplify[eq /. t → 0];

In[32]:= DSolve[% == 0, f3, y] // Simplify
```

Out[32]= $\left\{\left\{f3\to \mathrm{Function}\left[\{y\},\ -\frac{1}{6\,R1^5}(-1+R1)\left(5-4\,R1+8\,R2-4\,R1\,R2+2\,St-3\,R1\,St+R1^2\,St\right)y-\frac{1}{18\,R1^4}(-1+R1)\left(-6+12\,R2+5\,St-4\,R1\,St+8\,R2\,St-4\,R1\,R2\,St+2\,St^2-3\,R1\,St^2+R1^2\,St^2\right)y^2-\frac{1}{18\,R1^3}\left(-1+7\,R2-10\,R2^2+9\,R1\,R3+2\,St-2\,R1\,St-R2\,St+R1\,R2\,St\right)y^3-\frac{C[1]}{y}+C[2]\right]\right\}\right\}$

```
In[33]:= f3[y] /. %[[1]] /. C[1] → 0 // Simplify;

In[34]:= f3[y_] = %;

In[35]:= {bc1, bc2} /. t → 0 // Simplify;

In[36]:= Solve[% == 0, {C[2], s3}] // Simplify;
```

```
In[37]:= s3 = (s3 /. %[[1]]); f3[y_] = (Simplify[f3[y] /. %[[1]]]);

In[38]:= ss = Series[s[t], {t, 0, 3}] // Simplify
```

$$\text{Out[38]=}\ \left(-1+\frac{1}{R1}\right)t-\frac{1}{6\,R1^3}\left((-1+R1)\,(-1-4\,R2+(-1+R1)\,St)\right)t^2+\frac{1}{54\,R1^5}(-1+R1)\left(-4-46\,R2^2-9\,St-5\,St^2+2\,R1^3\,St^2-R2\,(23+27\,St)-R1^2\,(27\,R3+St\,(5+11\,R2+9\,St))+R1\,\left(3+30\,R2^2+27\,R3+14\,St+12\,St^2+R2\,(15+38\,St)\right)\right)t^3+O[t]^4$$

```
In[39]:= (* We need s'(t)≥0 and hence R1<1 *)

In[40]:= (* As R1→1 we then need s''(t)≥0 *)

In[41]:= Simplify[D[ss, t, t] /. t → 0]
```

$$\text{Out[41]=}\ -\frac{1}{3\,R1^3}(-1+R1)\,(-1-4\,R2+(-1+R1)\,St)$$

```
In[42]:= Simplify[% / (1 - R1)] /. R1 → 1
```

$$\text{Out[42]=}\ \frac{1}{3}\,(-1-4\,R2)$$

```
In[43]:= (* Therefore as R1→1 we need R2≥-1/4 *)

In[44]:= (* As R1→1 and R2→-1/4, we need s'''[0]≥0 *)

In[45]:= Simplify[D[ss, t, t, t] /. t → 0]
```

$$\text{Out[45]=}\ \frac{1}{9\,R1^5}(-1+R1)\left(-4-46\,R2^2-9\,St-5\,St^2+2\,R1^3\,St^2-R2\,(23+27\,St)-R1^2\,(27\,R3+St\,(5+11\,R2+9\,St))+R1\,\left(3+30\,R2^2+27\,R3+14\,St+12\,St^2+R2\,(15+38\,St)\right)\right)$$

```
In[46]:= (* R3 must exceed the zero of this coefficient *)

In[47]:= (* Let R2→-1/4 and R1→1 carefully! *)

In[48]:= Solve[Evaluate[D[ss, t, t, t] /. t → 0] == 0, R3] // Simplify
```

$$\text{Out[48]=}\ \left\{\left\{R3\to-\frac{1}{27\,(-1+R1)\,R1}\left(4+46\,R2^2+9\,St+5\,St^2-2\,R1^3\,St^2+R1^2\,St\,(5+11\,R2+9\,St)+R2\,(23+27\,St)-R1\,\left(3+30\,R2^2+14\,St+12\,St^2+R2\,(15+38\,St)\right)\right)\right\}\right\}$$

```
In[49]:= FullSimplify[(R3 /. %[[1]]) /. R2 → -1 / 4]
```

$$\text{Out[49]=}\ \frac{9+2\,(-1+R1)\,St\,(-9+4\,(-5+2\,R1)\,St)}{216\,R1}$$

```
In[50]:= FullSimplify[% /. R1 → 1]
```

$$\text{Out[50]=}\ \frac{1}{24}$$

```
In[51]:= (* This gives us the required maximal radius profile
```

$$R_m(z)\ \sim\ z-\frac{z^2}{4}+\frac{z^3}{24}+\ \ldots\ \text{*)}$$

6/11



BCs on free surface $r=R(z,t)$ are

① Kinematic BC $\boxed{u = \frac{\partial R}{\partial t} + w\frac{\partial R}{\partial z}}$ at $r=R(z,t)$

② Dynamic BC $\underline{\underline{\sigma}} \cdot \underset{\sim}{n} = \underline{0}$ where $\underset{\sim}{n} = \dfrac{\underset{\sim}{e}_r - \frac{\partial R}{\partial z}\underset{\sim}{e}_z}{\sqrt{1+\left(\frac{\partial R}{\partial z}\right)^2}}$

$$\therefore \boxed{\begin{aligned} \sigma_{rr} &= \frac{\partial R}{\partial z}\sigma_{rz} \\ \sigma_{rz} &= \frac{\partial R}{\partial z}\sigma_{zz} \end{aligned}} \quad \text{at } r=R(z,t)$$

Given $\frac{\partial}{\partial r}(ru) + \frac{\partial}{\partial z}(rw) = 0$

$$\therefore \left[ru\right]_{r=0}^{r=R} + \int_0^R \frac{\partial}{\partial z}(rw)\,dr = 0 \qquad \left(\begin{matrix}\text{NB } ru \to 0 \\ \text{as } r\to 0\end{matrix}\right)$$

$$\therefore R\left(\frac{\partial R}{\partial t} + w\frac{\partial R}{\partial z}\right)_{r=R} + \int_0^R \frac{\partial}{\partial z}(rw)\,dr = 0$$

$\therefore$ (by Leibniz)

$$R\frac{\partial R}{\partial t} + \frac{\partial}{\partial z}\int_0^R wr\,dr = 0$$

$$\therefore \boxed{\frac{\partial A}{\partial t} + \frac{\partial}{\partial z}(A\bar{w}) = 0}$$ as required.

similarly from $\frac{\partial}{\partial r}(r\sigma_{rz}) + \frac{\partial}{\partial z}(r\sigma_{zz}) = 0$

$$\left[r\sigma_{rz}\right]_0^R + \int_0^R \frac{\partial}{\partial z}(r\sigma_{zz})\,dr = 0$$

$$\therefore R\frac{\partial R}{\partial z}\sigma_{zz}\Big|_{r=R} + \int_0^R \frac{\partial}{\partial z}(r\sigma_{zz})\,dr = 0$$

$$\therefore \frac{\partial}{\partial z}\int_0^R \sigma_{zz}\, r\, dr = 0 \qquad \text{(Leibnitz)}$$

$$\therefore \boxed{\frac{\partial}{\partial z}(A\bar{\sigma}_{zz}) = 0}$$

Now non-dimensionalize given equations & BCs:

$u = \varepsilon U\tilde{u}$, $w = U\tilde{w}$, $z = L\tilde{z}$, $r = \varepsilon L\tilde{r}$, $t = \frac{L}{U}\tilde{t}$

$p = \frac{\mu U}{L}\tilde{p}$, $\sigma_{zz} = \frac{\mu U}{L}\tilde{\sigma}_{zz}$, $\sigma_{rz} = \frac{\varepsilon\mu U}{L}\tilde{\sigma}_{rz}$

$\sigma_{rr} = \frac{\varepsilon^2\mu U}{L}\tilde{\sigma}_{rr}$, $\sigma_{\theta\theta} = \frac{\varepsilon^2\mu U}{L}\tilde{\sigma}_{\theta\theta}$

Dimensionless problem

$$\frac{1}{r}\frac{\partial}{\partial r}(ru) + \frac{\partial w}{\partial z} = 0$$

$$\frac{1}{r}\frac{\partial}{\partial r}(r\sigma_{rr}) - \frac{\sigma_{\theta\theta}}{r} + \frac{\partial \sigma_{rz}}{\partial z} = 0$$

$$\frac{1}{r}\frac{\partial}{\partial r}(r\sigma_{rz}) + \frac{\partial \sigma_{zz}}{\partial z} = 0$$

constitutive relations

$$\varepsilon^2 \sigma_{rr} = -p + 2\frac{\partial u}{\partial r}, \qquad \sigma_{zz} = -p + 2\frac{\partial w}{\partial z}$$

$$\varepsilon^2 \sigma_{\theta\theta} = -p + \frac{2u}{r}, \qquad \varepsilon^2 \sigma_{rz} = \varepsilon^2 \frac{\partial u}{\partial z} + \frac{\partial w}{\partial r}$$

B/c Boundedness as $r \to 0$ and

$$\left.\begin{aligned} u &= \frac{\partial R}{\partial t} + w\frac{\partial R}{\partial z} \\ \sigma_{rr} &= \frac{\partial R}{\partial z}\sigma_{rz}, \quad \sigma_{rz} = \frac{\partial R}{\partial z}\sigma_{zz} \end{aligned}\right\} \text{at } r = R(z,t).$$

Now let $\varepsilon \to 0$: leading order equations

$$\frac{\partial w}{\partial r} = 0 \Rightarrow \boxed{w = W(z,t)}$$ extensional flow.

Then $$ru = -\frac{r^2}{2}\frac{\partial W}{\partial z}$$ (integration constant $= 0$)

$$\boxed{u = -\frac{r}{2}\frac{\partial W}{\partial z}}$$

then
$$\boxed{p = 2\frac{\partial u}{\partial r} = \frac{2u}{r} = -\frac{\partial w}{\partial z}}$$

& hence
$$\boxed{\sigma_{zz} = 3\frac{\partial w}{\partial z}}$$

So integrated eq^{ns} obtained above lead to Trouton model:

$$\boxed{\begin{aligned} &\frac{\partial A}{\partial t} + \frac{\partial}{\partial z}(Aw) = 0 \\ &\frac{\partial}{\partial z}\left(3A\frac{\partial w}{\partial z}\right) = 0 \end{aligned}}$$

7. (2.74)

$$\boxed{\begin{aligned} &\frac{d}{dz}\left(\tilde{A}+\tilde{w}\right) + \sigma e^{-mz}\tilde{A} = 0 \\ &\frac{d\tilde{w}}{dz} + m\left(\tilde{A}+\tilde{w}\right) = m\tilde{F} \\ &\tilde{A}(0) = \tilde{w}(0) = \tilde{w}(1) = 0 \end{aligned}}$$

$$\therefore \quad \tilde{A} = \tilde{F} - \tilde{w} - \frac{1}{m}\frac{d\tilde{w}}{dz}$$

& then
$$-\frac{1}{m}\frac{d^2\tilde{w}}{dz^2} + \sigma e^{-mz}\left[\tilde{F} - \tilde{w} - \frac{1}{m}\frac{d\tilde{w}}{dz}\right] = 0$$

i.e.
$$\frac{d^2\tilde{w}}{dz^2} + \sigma e^{-mz}\left[\frac{d\tilde{w}}{dz} + m\left(\tilde{w} - \tilde{F}\right)\right] = 0$$

subject to
$$\begin{cases} \tilde{w}(0) = \tilde{w}(1) = 0 \\ \dfrac{d\tilde{w}}{dz}(0) = m\tilde{F} \end{cases}$$

Now make suggested substitution:

$$\tilde{w} = \tilde{F}(1-\phi), \qquad x = \frac{\sigma}{m}e^{-mz}$$

$$\Rightarrow \quad \frac{d}{dz} = -mx\frac{d}{dx} \; ; \quad \frac{d\tilde{w}}{dz} = mx\tilde{F}\frac{d\phi}{dx}$$

$$m^2 x \frac{d}{dx}\left(x \frac{d\phi}{dx}\right) + mx\left[-mx\frac{d\phi}{dx} + m\phi\right] = 0$$

$$\therefore \; x^2 \frac{d^2\phi}{dx^2} + x\frac{d\phi}{dx} - x^2\frac{d\phi}{dx} + x\phi = 0$$

$$\therefore \; \boxed{\frac{d^2\phi}{dx^2} + \left(\frac{1}{x} - 1\right)\frac{d\phi}{dx} + \frac{1}{x}\phi = 0}$$

BCs:

$$\boxed{\begin{array}{ll} \phi = 1, \quad x\dfrac{d\phi}{dx} = 1, & x = \lambda = \sigma/m \\ \phi = 1, & x = \dfrac{\sigma}{m}e^{-m} = \lambda/D \end{array}}$$

Try $\phi = x - 1$; $\phi' = 1$, $\phi'' = 0$

so LHS $= 0 + \left(\frac{1}{x} - 1\right) + \frac{1}{x}(x-1) = 0$ ✓

So now try $\boxed{\phi(x) = (x-1)V(x)}$

Then $(x-1)V'' + 2V' + \left(\frac{1}{x} - 1\right)\left((x-1)V' + V\right) + \left(\frac{x-1}{x}\right)V = 0$

$$\Rightarrow x(x-1)V'' + \left[2x - x^2 + 2x - 1\right]V' = 0$$

$$\Rightarrow \frac{V''}{V'} = \frac{x^2 - 4x + 1}{x(x-1)} = 1 - \frac{1}{x} - \frac{2}{x-1}$$

$$\therefore \quad \boxed{V' = \frac{C.\,e^{x}}{x(x-1)^2}}$$

BCs for V:

$$\boxed{V(\lambda) = \frac{1}{\lambda-1}, \quad V'(\lambda) = -\frac{1}{\lambda(\lambda-1)^2}, \quad V(\lambda/D) = \frac{1}{\lambda/D-1}}$$

Hence evaluate C

$$V'(x) = -\frac{e^{x-\lambda}}{x(x-1)^2} \quad \& \quad \int_{\lambda/D}^{\lambda} V'(x)\,dx = V(\lambda) - V(\lambda/D)$$

$$\cdots \Rightarrow \quad \boxed{\int_{\lambda/D}^{\lambda} \frac{e^{x-\lambda}\,dx}{x(x-1)^2} = \frac{1}{\lambda/D-1} - \frac{1}{\lambda-1}}$$

If $\lambda = i\omega$ where $\omega \in \mathbb{R}^+$ (wlog), then we integrate along the imaginary axis: let $x = i\xi$



$$\boxed{\int_{\omega/D}^{\omega} \frac{e^{i\xi}\,d\xi}{\xi(1-i\xi)^2} + e^{i\omega}\left[\frac{1}{1-i\omega/D} - \frac{1}{1-i\omega}\right] = 0}$$

```
In[1]:= Integrate[Exp[I x] / x / (1 - I x) ^2, x]
```

Out[1]= $\dfrac{i\left(e^{i x}+(1-i x) \text{ExpIntegralEi}[i x]\right)}{i+x}$

```
In[2]:= (% /. x → w) - (% /. x → w / Dr) + Exp[I w] (1 / (1 - I w / Dr) - 1 / (1 - I w)) // FullSimplify
```

Out[2]= $\dfrac{\text{Dr}\left(e^{i w}-e^{\frac{i w}{\text{Dr}}}\right)+(\text{Dr}-i w)\left(\text{ExpIntegralEi}[i w]-\text{ExpIntegralEi}\left[\frac{i w}{\text{Dr}}\right]\right)}{\text{Dr}-i w}$

```
In[3]:= F[w_, Dr_] = %;

In[4]:= ContourPlot[Evaluate[{Re[F[w, Dr]] == 0, Im[F[w, Dr]] == 0}],
          {w, 0, 20}, {Dr, 1.5, 100}, Contours → {0}]
```

Out[4]=




```
In[5]:= FindRoot[Evaluate[{Re[F[w, Dr]], Im[F[w, Dr]]} == 0], {{w, 4.5}, {Dr, 20}}]
```

Out[5]= {w → 4.66015, Dr → 20.218}

§11 $\quad A_0(x) = 1 + \alpha\left(x - \frac{1}{2}\right)^2, \quad \alpha > 0$

So
$$\begin{cases} A = 1 - f(\tau) + \alpha\left(s - \frac{1}{2}\right)^2 \\ z = s + f(\tau)\displaystyle\int_0^s \frac{ds'}{1 - f(\tau) + \alpha\left(s' - \frac{1}{2}\right)^2} \end{cases}$$

Note critical value $f = 1$ where $A \to 0$ at $s = \frac{1}{2}$ and $z \to \infty$

Calculate integral:

$$z = s + \frac{f}{\sqrt{\alpha(1-f)}}\left[\tan^{-1}\left(\frac{1}{2}\sqrt{\frac{\alpha}{1-f}}\right) + \tan^{-1}\left(\left(s - \frac{1}{2}\right)\sqrt{\frac{\alpha}{1-f}}\right)\right]$$

$$\therefore \boxed{S = 1 + \frac{2f}{\sqrt{\alpha(1-f)}}\tan^{-1}\left(\frac{1}{2}\sqrt{\frac{\alpha}{1-f}}\right)}$$

As $f \to 1$, $\quad S \sim 1 + \frac{2}{\sqrt{\alpha(1-f)}} \cdot \frac{\pi}{2}$

$$\Rightarrow \boxed{f \sim 1 - \frac{\pi^2}{\alpha S^2} \quad \text{as } S \to \infty}$$

In[1]:= `1/(1 - f + a (x - 1/2)^2)`

Out[1]= $$\frac{1}{1-f+a\left(-\frac{1}{2}+x\right)^2}$$

In[2]:= `Integrate[%, x]`

Out[2]= $$\frac{\text{ArcTan}\left[\frac{\sqrt{a}\left(-\frac{1}{2}+x\right)}{\sqrt{1-f}}\right]}{\sqrt{a}\sqrt{1-f}}$$

In[3]:= `(% /. x → ze) - (% /. x → 0) // FullSimplify`

Out[3]= $$\frac{\text{ArcCot}\left[\frac{2\sqrt{1-f}}{\sqrt{a}}\right]+\text{ArcTan}\left[\frac{\sqrt{a}\left(-\frac{1}{2}+ze\right)}{\sqrt{1-f}}\right]}{\sqrt{a}\sqrt{1-f}}$$

In[4]:= `z = ze + f %`

Out[4]= $$ze+\frac{f\left(\text{ArcCot}\left[\frac{2\sqrt{1-f}}{\sqrt{a}}\right]+\text{ArcTan}\left[\frac{\sqrt{a}\left(-\frac{1}{2}+ze\right)}{\sqrt{1-f}}\right]\right)}{\sqrt{a}\sqrt{1-f}}$$

In[5]:=
```
ParametricPlot[
 Evaluate[Table[{z, 1 - f + a (ze - 1/2)^2} /. a → 0.5, {f, 0, 0.9, 0.1}]],
 {ze, 0, 1}, PlotRange → All, AspectRatio → 1/GoldenRatio,
 AxesOrigin → {0, 0}, AxesLabel → {"z", "A"}, PlotLabel → "Here α = 0.5",
 PlotLegends → Join[{f == 0}, Table[f, {f, 0.1, 0.9, 0.1}]]]
```

Out[5]=



In[6]:= `z /. ze → 1 // FullSimplify`

Out[6]= $$1+\frac{2\,f\,\text{ArcCot}\left[\frac{2\sqrt{1-f}}{\sqrt{a}}\right]}{\sqrt{a}\sqrt{1-f}}$$

In[7]:= `s = %`

Out[7]= $$1+\frac{2\,f\,\text{ArcCot}\left[\frac{2\sqrt{1-f}}{\sqrt{a}}\right]}{\sqrt{a}\sqrt{1-f}}$$

In[8]:=
```
p1 = ParametricPlot[Evaluate[{s, f}] /. a → 0.5, {f, 0, 0.99},
  AspectRatio → 1 / GoldenRatio, PlotRange → {{0, 20}, {0, 1}},
  AxesLabel → {"s", "f"}, PlotLabel → "α = 0.5"]
```

Out[8]=



In[9]:=
```
s - ss /. f → 1 - Pi^2 / a / ss^2 - f3 / ss^3 - f4 / ss^4 // Simplify // PowerExpand
```

Out[9]=
$$1 - ss - \frac{2\left(\pi^2\,ss^2 + a\left(f4 + f3\,ss - ss^4\right)\right)\,\mathrm{ArcCot}\left[\frac{2\sqrt{a\,f4 + a\,f3\,ss + \pi^2\,ss^2}}{a\,ss^2}\right]}{a\,ss^2\,\sqrt{a\,f4 + a\,f3\,ss + \pi^2\,ss^2}}$$

In[10]:=
```
Assuming[a > 0 && f2 > 0 && ss > 0, Series[%, {ss, Infinity, 1}] // Simplify]
```

Out[10]=
$$\left(1 - \frac{4}{a} - \frac{a\,f3}{2\,\pi^2}\right) + \frac{\frac{3\,a^2\,f3^2}{8\,\pi^4} - \frac{a\,f4}{2\,\pi^2} - \frac{\pi^2}{a}}{ss} + O\left[\frac{1}{ss}\right]^2$$

In[11]:=
```
Solve[% == 0, {f3, f4}] // Simplify
```

Out[11]=
$$\left\{\left\{f3 \to \frac{2\,(-4 + a)\,\pi^2}{a^2},\ f4 \to \frac{\pi^2\left(48 + 3\,a^2 - 2\,a\left(12 + \pi^2\right)\right)}{a^3}\right\}\right\}$$

In[12]:=
```
1 - Pi^2 / a / ss^2 - f3 / ss^3 - f4 / ss^4 /. %[[1]]
```

Out[12]=
$$1 - \frac{\pi^2\left(48 + 3\,a^2 - 2\,a\left(12 + \pi^2\right)\right)}{a^3\,ss^4} - \frac{2\,(-4 + a)\,\pi^2}{a^2\,ss^3} - \frac{\pi^2}{a\,ss^2}$$

In[13]:=
```
Plot[Evaluate[% /. a → 0.5], {ss, 10, 20}, PlotStyle → Dashing[{0.01, 0.01}]]
```

Out[13]=



In[14]:= `Show[p1, %, PlotLabel → "Comparison with asymptotics"]`

Out[14]=



9/ Write Trouton model with inertia in the form:

$$\frac{DA}{Dt} = -A\frac{\partial w}{\partial z}, \qquad R A \frac{Dw}{Dt} = \frac{\partial}{\partial z}\left(3A\frac{\partial w}{\partial z}\right)$$

where $D/Dt = \partial/\partial t + w\,\partial/\partial z$

Now $$\frac{\partial}{\partial z}\left(A\frac{\partial w}{\partial z}\right) = -\frac{\partial}{\partial z}\left[\frac{\partial A}{\partial t} + w\frac{\partial A}{\partial z}\right]$$

$$= -\frac{D}{Dt}\left(\frac{\partial A}{\partial z}\right) - \frac{\partial w}{\partial z}\frac{\partial A}{\partial z} = \frac{R}{3}A\frac{Dw}{Dt}$$

$$\therefore \frac{R}{3}A\frac{Dw}{Dt} + \frac{D}{Dt}\left(\frac{\partial A}{\partial z}\right) - \frac{\partial A}{\partial z}\cdot\frac{1}{A}\frac{DA}{Dt} = 0$$

Divide by A & use product rule:

$$\boxed{\frac{D}{Dt}\left(\frac{R}{3}w + \frac{1}{A}\frac{\partial A}{\partial z}\right) = 0}$$

Initially at rest & uniform: $w=0$, $A=1$ at $t=0$

For $t>0$, hold $z=0$ fixed & draw off at speed 1 at $z=1$:



From above eqⁿ we get

$$\frac{R}{3}w + \frac{1}{A}\frac{\partial A}{\partial z} = \text{const.} \quad \text{along characteristics} \quad dz/dt = w.$$

. From ICs, this constant is zero. So

$$\boxed{w \equiv -\frac{3}{RA}\frac{\partial A}{\partial z}}$$

& conservation of mass $\Rightarrow$ $\boxed{\frac{R}{3}\frac{\partial A}{\partial t} = \frac{\partial^2 A}{\partial z^2}}$ heat equation!

BCs $w(0)=0$, $w(1)=1$

$$\Rightarrow \boxed{\begin{array}{ll} \frac{\partial A}{\partial t} = 0 & \text{at } z=0 \\ \frac{\partial A}{\partial z} + \frac{R}{3}A = 0 & \text{at } z=1 \end{array}}$$

& $\boxed{A = 1 \quad \text{at } t=0}$

Just separate variables: $A(z,t) = \cos(kz)e^{-3k^2t/R}$

works $\forall k$ satisfying $-k\sin k + \frac{R}{3}\cos k = 0$

i.e. $\boxed{k\tan k = R/3}$

Then by linear superposition:

$$\boxed{A(z,t) = \sum_{n=1}^{\infty} c_n \cos(k_n z)e^{-3k_n^2t/R}}$$

where $\quad C_n \int_0^1 \cos^2(k_n z)\, dz = \int_0^1 \cos(k_n z)\, dz$

by using initial conditions & orthogonality:

$$\left(\int_0^1 \cos(k_n z) \cos(k_m z)\, dz = 0 \quad \text{for } n \neq m \right)$$

Evaluate integrals: $\quad C_n = \dfrac{2 \sin(k_n)}{k_n + \sin(k_n)\cos(k_n)}$

In[1]:=
```
kn[n_, R_] :=
  (k /. FindRoot[k Sin[k] == R Cos[k] / 3, {k, (n - 1) Pi, (n - 1 / 2) Pi}][[1]])
```

In[2]:=
```
Integrate[Cos[k z], {z, 0, 1}] / Integrate[Cos[k z]^2, {z, 0, 1}] // FullSimplify
```

Out[2]=
$$\frac{2\,\mathrm{Sin}[k]}{k + \mathrm{Cos}[k]\,\mathrm{Sin}[k]}$$

In[3]:=
```
A[R_, m_, z_, t_] := Sum[(2 Sin[kn[n, R]]) / (kn[n, R] + Cos[kn[n, R]] Sin[kn[n, R]])
    Cos[kn[n, R] z] Exp[-3 kn[n, R]^2 t / R], {n, 1, m}]
```

In[4]:=
```
pl[Re_] :=
 Plot[Evaluate[Join[{1}, Table[Evaluate[A[Re, 20, z, t]], {t, 0.5, 3, 0.5}]]],
  {z, 0, 1}, PlotLabel → R == Re, AxesLabel → {"z", "A"},
  PlotLegends → Join[{t == 0}, Table[t, {t, 0.5, 3, 0.5}]], PlotRange → {0, 1}]
```

In[5]:=
```
pl[1.0]
```

Out[5]=


In[6]:=
```
pl[0.1]
```

Out[6]=


In[7]:=
```
(* When R is small, A is approximately spatially uniform.*)
```

33

In[8]:= `pl[100]`

Out[8]=



In[9]:= `(* When R is large, there is a boundary layer near z=1. *)`